
Checking Sufficient Completeness

We need methods to check that an equational theory (Σ, E) is
sufficiently complete. For arbitrary equational theories sufficient
completeness is in general undecidable. This is not so bad: it just
means that we may have to do some inductive theorem proving.

Sufficient completeness is decidable for a very broad class of
order-sorted theories, namely, unconditional theories of the form
(Σ, E ∪B) with B a set of axioms for operators allowing any
combination of associativity and/or commutativity and/or identity,
except associativity without commutativity, and E⃗: (i) left-linear;
(ii) sort-decreasing; and (iii) terminating.

Checking Sufficient Completeness (II)

Furthermore, even for cases satisfying the above requirements
(i)–(iii), but where B includes operators that are only associative,
or associative and identity, sufficient completeness, although
undecidable in theory, becomes decidable in practice for many
specifications of interest using specialized heuristic algorithms.

Left-linearity (i) means that if t = t′ ∈ E, then t has no repeated
variables, i.e., if x ∈ vars(t), then x occurs at a single position p in
t. This fails, e.g., for the idempotency equation x ∪ x = x, where x

occurs at positions 1 and 2 in x ∪ x. Properties (ii)–(iii) (modulo
B) we are alredy familiar with.

Tree Automata

For equational theories satisfying the above requirements (i)–(iii)
we can use decidability results from tree automata theory to cast
the sufficient completeness problem into a tree automata problem
and decide the problem that way.

An ordinary finite-state automaton A has a finite set Q of states
and accepts strings of inputs when they lead the automaton to a
subset Q0 ⊆ Q of accepting states. The language L(A) of the
automaton is then the set of all strings accepted by A. Such
languages are called regular languages and have nice decidablity
results: they are closed under Boolean operations (we can construct
automata for each such operation); and we can decide whether
L(A) is empty.

Tree Automata (II)

All this is generalized by finite-state tree automata, which accept
terms in an unsorted term algebra TΣ instead of just strings. A
tree automaton (TA) is a tuple A = (Σ, Q,Q0, R) with Σ an
unsorted signature, Q a set of extra constants not in Σ called
states, Q0 ⊆ Q a subset of accepting states, and R a set of
transition rules, which can be of two forms:

• f(q1, . . . , qn) → q, with q1, . . . , qn, q ∈ Q, f ∈ Σ and
f(q1, . . . , qn) ∈ TΣ(Q) (for n = 0, f will be a constant, and the
rule becomes f → q)

• q → q′, with q, q′ ∈ Q (epsilon transition)

Tree Automata (III)

Notice that we can view the transition rules R as ground rewrite
rules and can use them to rewrite terms in the term algebra TΣ(Q).
I.e., the TA specifies a ground TRS (Σ(Q), R) (plus the subset
Q0 ⊆ Q). Notice also that we have an inclusion TΣ ⊆ TΣ(Q). We
then define the language L(A) as the subset L(A) ⊆ TΣ of those
t ∈ TΣ such that there is a q ∈ Q0 such that t →∗

R q. A subset
L ⊆ TΣ is called regular iff there is a finite-state tree automaton A
such that L = L(A).

Tree automata have the same decidablity results as string
automata: they are closed under Boolean operations (we can
construct automata for each such operation); and we can decide
whether L(A) is empty.

Tree Automata as Order-Sorted Signatures

Automata are labeled graphs. Tree automata are labeled
multigraphs, actually, order-sorted signatures. Any tree automaton
A = (Σ, Q,Q0, R), with Σ = ({s}, F,Σ), is the order-sorted
signature: ΣA = ((Q,<), F,ΣA), where,

q < q′ ⇔ q →+
Rϵ

q′

where Rϵ are the epsilon transitions (w.l.o.g. may assume Rϵ

terminating). And for each q1, . . . , qn, q ∈ Q we have:

f(q1, . . . , qn) → q in R ⇔ f : q1, . . . , qn → q in ΣA

If the accepting states are Q0 = {q1, . . . , qk}, then the regular
language L(A) is:

L(A) = TΣA,q1
∪ . . . ∪ TΣA,qk

.

This is because, for each q ∈ Q, t ∈ TΣA,q
iff t →∗

R q

Tree Automata for Sufficient Completeness

For checking sufficient completeness, the key observation is that,
for theories (Σ, E ∪B) satisfying conditions (i)–(iii), the following
sets of ground Σ-terms are regular sets:

• the set Ds of terms of sort s having a defined symbol on top
and constructor terms as arguments;

• the set Cs of constructor terms of sort s; and

• the set Red of terms reducible by the equations E (modulo B),
i.e., terms not in normal form.

Under conditions (i)–(iii) (Σ, E ∪B) is sufficiently complete iff for
each sort s we have Ds − (Red ∪ Cs) = ∅, which can be decided by
deciding emptiness of the corresponding tree automaton.

8



Tree Automata Modulo B

In general, we need to consider tree automata modulo B (TA/B),
that is, tuples A = (Σ, B,Q,Q0, R), where (Σ, Q,Q0, R) is an
ordinary TA, and B is a set of equational Σ-axioms such as
associativity, commutativity, and identity of some symbols. I.e., the
TA/B specifies a ground rewrite theory (Σ(Q), B,R) (plus the
subset Q0 ⊆ Q). The language of A is then a subset L(A) ⊆ TΣ,
now defined by rewriting with R modulo B. That is, L(A) ⊆ TΣ is
the set of those t ∈ TΣ such that there is a q ∈ Q0 such that
t →∗

R/B q.

Hendrix, Ohsaki, and Viswanathan extend the tree automata
decidablity results to the modulo B case, for B any combination of
associativity and/or commutativity and/or identity, except
associativity without commutativity.

Tree Automata Modulo B with A and not C Axioms

Even in the case when B contains axioms for a binary operator f

that is associative (A) or AU , but not commutative, for which some
tree automata questions like emptiness become undecidable, the
sufficient completeness problem can still be decided in practice for
many cases of interest by specialized heuristic algorithms (Hendrix,
Ohsaki, and Viswanathan, Proc. RTA 2006, Springer LNCS).

All this means that in practice we can decide the sufficient
completeness of most left-linear order-sorted specifications of
interest.

10



An Example

To see how the desired tree automata needed to decide sufficient
completeness can be built, we can use a simple example, our usual
unsorted specification for addition for the Peano natural numbers
with a single sort Nat, with 0 and s as constructors, and with
equations x+ 0 = x and x+ s(y) = s(x+ y). This specification
satisfies conditions (i)–(iii), since it is left-linear, confluent,
sort-decreasing, and terminating.

To recognize each of the regular sets Red, DNat, and CNat we need
three tree automata ARed, ADNat

, and ACNat
.

An Example (II)

The tree automata ARed, ADNat
and ACtor have the same

signature Σ (that of the natural numbers), set of states
Q = {Nat,Red,DNat,Ctor ,Zero,NzCtor}, and transitions R:

• (ϵ-transitions): Ctor → Nat, Red → DNat, DNat → Nat,

Zero → Ctor , NzCtor → Ctor .

• (Nat transitions) s(Nat) → Nat, Nat+Nat → Nat.

• (constructor transitions): 0 → Zero, s(Ctor) → NzCtor .

• (defined function transition): Ctor + Ctor → DNat.

• (reducibility transitions): Ctor + Zero → Red,
Ctor +NzCtor → Red.

An Example (III)

ARed, ADNat
and ACtor only differ in their respective accepting

state: Red, DNat, and Ctor . Since their: (i) signature, (ii) set of
states and (iii) transitions are the same, the shared part (i)–(iii) is
just the order sorted signature:

sorts Nat Red, D-Nat Ctor Zero NzCtor .
subsorts Zero NzCtor < Ctor < Nat .
subsorts Red < D-Nat < Nat .
op s : Nat Nat -> Nat .
op s : Ctor -> NzCtor .
op 0 : -> Zero .
op _+_ : Nat Nat -> Nat .
op _+_ : Ctor Ctor -> D-Nat .
op _+_ : Ctor Zero -> Red .
op _+_ : Ctor NzCtor -> Red .

An Example (IV)

The point now is that each Boolean operation on regular tree
languages has a corresponding operation on their associated tree
automata. Therefore, out of the automata ARed, ADNat

, and ACtor

we can construct an automaton that recognizes the language
DNat − (Red ∪ Ctor). Let us call this automaton
ADNat−(Red∪Ctor). Now we know that under conditions (i)–(iii) our
specification is sufficiently complete iff DNat − (Red ∪ Ctor) = ∅.
Therefore, we can decide this property by testing ADNat−(Red∪Ctor)

for emptiness. If the test (as for this example) succeeds, we are
done. If it doesn’t, we get very useful counterexample terms,
showing us where sufficient completeness fails.

The Maude SCC Tool

The Maude Sufficient Completeness Checker (SCC) is a tool
developed by Joseph Hendrix at UIUC. It uses a library of tree
automata modulo B operations also developed by him, and reduces
the sufficient completeness problem of specification (Σ, E ∪B)

satisfying conditions (i)–(iii) to the emptiness problem for the tree
automaton ADs−(Red∪Cs) for each sort s in Σ. It outputs either
“success” or a set of counterexample terms.

Instructions to acces SCC can be found in the course web page. Its
use is essentially very simple. One: (1) loads the module
scc.maude; (2) loads the module to be checked, say FOO; (3) types
“select SCC-LOOP .” and “loop init-scc .” and (4) gives to
the SCC the command “(scc FOO .)”.

The Maude SCC Tool (II)

We can illustrate the use of the SCC with some examples already
encountered previously in the course. The module NATURAL was
already used above as an example to illustrate the tree automata
ARed, ADNat

and ACtor . The SCC tool will now check the
emptiness of ADNat−(Red∪Ctor).

fmod NATURAL is
sort Nat .
op 0 : -> Nat [ctor] .
op s : Nat -> Nat [ctor] .
op _+_ : Nat Nat -> Nat .
vars X Y : Nat .
eq X + 0 = X .
eq X + s(Y) = s(X + Y) .
endfm

The Maude SCC Tool (III)

Emptiness of ADNat−(Red∪Ctor). is successfully checked by SCC:
Maude> load scc .
Maude> in natural .
==========================================
fmod NATURAL
Maude> select SCC-LOOP .
Maude> loop init-scc .
Starting the Maude Sufficient Completeness Checker.
Maude> (scc NATURAL .)
Checking sufficient completeness of NATURAL ...
Warning: This module has equations that are not

left-linear. The sufficient completeness checker will
rename variables as needed to drop the non-linearity
conditions.

Success: NATURAL is sufficiently complete under the
assumption that it is weakly-normalizing, confluent,
and sort-decreasing.

The Maude SCC Tool (IV)

Consider the module

fmod MY-LIST is
protecting NAT .
sorts NzList List .
subsorts Nat < NzList < List .
op _;_ : List List -> List [assoc] .
op _;_ : NzList NzList -> NzList [assoc ctor] .
op nil : -> List [ctor] .
op rev : List -> List .
eq rev(nil) = nil .
eq rev(N:Nat) = N:Nat .
eq rev(N:Nat ; L:List) = rev(L:List) ; N:Nat .

endfm

The Maude SCC Tool (V)

when checked by the SCC gives us the counterexample
Maude> load scc
Maude> in mylist
==========================================
fmod MY-LIST
Maude> select SCC-LOOP .
Maude> loop init-scc .
Starting the Maude Sufficient Completeness Checker.
Maude> (scc MY-LIST .)
Checking sufficient completeness of MY-LIST ...
Warning: This module has equations that are not

left-linear. The sufficient completeness checker will
rename variables as needed to drop the non-linearity
conditions.

Failure: The term 0 ; nil is a counterexample as it is a
irreducible term with sort List in MY-LIST that does
not have sort List in the constructor subsignature.

The Maude SCC Tool (VI)

We can correct this problem revising our module:

fmod MY-LIST2 is
protecting NAT .
sorts NzList List .
subsorts Nat < NzList < List .
op _;_ : List List -> List [assoc] .
op _;_ : NzList NzList -> NzList [assoc ctor] .
op nil : -> List [ctor] .
op rev : List -> List .
eq rev(nil) = nil .
eq rev(N:Nat) = N:Nat .
eq rev(N:Nat ; L:List) = rev(L:List) ; N:Nat .
eq nil ; L:List = L:List .
eq L:List ; nil = L:List .

endfm

The Maude SCC Tool (VII)

which is now successfully checked by SCC:
Maude> load scc
Maude> in mylist2
==========================================
fmod MY-LIST2
Maude> select SCC-LOOP .
Maude> loop init-scc .
Starting the Maude Sufficient Completeness Checker.
Maude> (scc MY-LIST2 .)
Checking sufficient completeness of MY-LIST2 ...
Warning: This module has equations that are not

left-linear. The sufficient completeness checker will
rename variables as needed to drop the non-linearity
conditions.

Success: MY-LIST2 is sufficiently complete under the
assumption that it is weakly-normalizing, confluent,
and sort-decreasing.

